Smart Playlist in Itunes based on file location
I don't think so. They appear to use information from the iTunes database/Library so how could they know anything about some random folder in your computer?

You could though simulate it using either Folder Actions or launchd but neither is perfect.

Folder Actions only works for new files in a directory and it provides the list of files added to the program/script you will run so you know what's changed. 

launchd on the other hand will run your script/app whenever anything changes in the folder(s), you can monitor mupltiple, but it will not tell you which path changed or how - added or removed etc.

I would go with launchd personally. If it is just a single folder you could hardcode it into your script.

As for the script, try this...

set dpath to "/some/path/test/"
set qdpath to quoted form of dpath
set listname to "MyList"
set adpath to (POSIX file dpath as alias)


# Get audio files in selected folder 
#   -- alas no recursive search results without a lot more code... :(
#   -- or use mdfind or other command line tool using `do shell script...`
set afiles to {}
tell application "Finder"
    repeat with ext in {".mp3", ".m4a"}
        set l to (every file in adpath whose name contains ext)
        repeat with f in l
            set end of afiles to (f as alias)
        end repeat
    end repeat
end tell

tell application "iTunes"
    if not (user playlist listname exists) then
        make new user playlist with properties {name:listname}
    end if

    # Built a list of existing list tracks so we don't have duplicates...
    set itracks to {}
    try
        set itracks to (get location of every track in playlist listname)
    end try

    set view of window 1 to playlist listname

    # Add not already in playlist
    repeat with afile in afiles
        if afile is not in itracks then
            try
                set tid to (add afile to playlist listname)
            end try
        end if
    end repeat

    # Remove any tracks from the itunes playlist not in the folder
    # and from the iTunes database else they still show up as invalid tracks...
    set iItems to (every track in playlist listname)
    set iLib to library playlist named "Library"
    repeat with iItem in iItems
        set loc to location of iItem
        if loc is not in afiles then
            set pid to persistent ID of iItem

            log "Removing: " &amp; loc
            try
                # Thanks Doug Adams, it was driving me nuts....
                delete (some file track of library playlist 1 whose persistent ID is pid)
            end try
            delay 1
        end if
    end repeat
end tell


It is actually fun to watch it react to adding and removing files from the folder. Enjoy.... :)

Finally I'd reccommend you check these URLs.

www.macissues.com/2015/02/02/how-to-use-launchagents-to-monitor-folder-contents-in-os-x/
apple.stackexchange.com/questions/6658/